

3+1 slicing of Kerr spacetime

This worksheet demonstrates a few capabilities of [SageManifolds](#) (version 1.0, as included in SageMath 7.5) in computations regarding the 3+1 slicing of Kerr spacetime.

Click [here](#) to download the worksheet file (ipynb format). To run it, you must start SageMath within the Jupyter notebook, via the command `sage -n jupyter`

NB: a version of SageMath at least equal to 7.5 is required to run this worksheet:

```
In [1]: version()
```

```
Out[1]: 'SageMath version 7.5.1, Release Date: 2017-01-15'
```

First we set up the notebook to display mathematical objects using LaTeX rendering:

```
In [2]: %display latex
```

Since some computations are quite long, we ask for running them in parallel on 8 cores:

```
In [3]: Parallelism().set(nproc=8)
```

Spacelike hypersurface

We consider some hypersurface Σ of a spacelike foliation $(\Sigma_t)_{t \in \mathbb{R}}$ of Kerr spacetime; we declare Σ_t as a 3-dimensional manifold:

```
In [4]: Sig = Manifold(3, 'Sigma', r'\Sigma', start_index=1)
print(Sig)
```

```
3-dimensional differentiable manifold Sigma
```

On Σ , we consider the "rational-polynomial" coordinates (r, y, ϕ) inherited from the standard Boyer-Lindquist coordinates (t, r, θ, ϕ) of Kerr spacetime, via $y = \cos \theta$.

```
In [5]: X.<r,y,ph> = Sig.chart(r'r:(1,+oo) y:(-1,1) ph:(0,2*pi):\phi')
print(X) ; X
```

```
Chart (Sigma, (r, y, ph))
```

```
Out[5]: ( $\Sigma, (r, y, \phi)$ )
```

Riemannian metric on Σ

First the two Kerr parameters:

```
In [6]: var('m, a', domain='real')
assume(m>0)
assume(a>0)
assumptions()
```

```
Out[6]: [r is real, r > 1, y is real, y > (-1), y < 1, ph is real, phi > 0, phi < 2 pi,
         m is real, a is real, m > 0, a > 0]
```

For dealing with extreme Kerr, the following must be uncommented:

```
In [7]: # m = 1 ; a = 1
```

Some shortcut notations:

```
In [8]: rho2 = r^2 + a^2*y^2
Del = r^2 - 2*m*r + a^2
AA2 = rho2*(r^2 + a^2) + 2*a^2*m*r*(1-y^2)
BB2 = r^2 + a^2 + 2*a^2*m*r*(1-y^2)/rho2
```

The metric γ induced by the spacetime metric g on Σ :

```
In [9]: gam = Sig.riemannian_metric('gam', latex_name=r'\gamma')
gam[1,1] = rho2/Del
gam[2,2] = rho2/(1-y^2)
gam[3,3] = BB2*(1-y^2)
gam.display()
```

Out[9]:

$$\gamma = \left(\frac{a^2 y^2 + r^2}{a^2 - 2mr + r^2} \right) dr \otimes dr + \left(-\frac{a^2 y^2 + r^2}{y^2 - 1} \right) dy \otimes dy$$

$$+ \left(\frac{2(y^2 - 1)a^2 mr}{a^2 y^2 + r^2} - a^2 - r^2 \right) (y^2 - 1) d\phi \otimes d\phi$$

A matrix view of the components w.r.t. coordinates (r, y, ϕ) :

```
In [10]: gam[:]
```

Out[10]:

$$\begin{pmatrix} \frac{a^2 y^2 + r^2}{a^2 - 2mr + r^2} & 0 & 0 \\ 0 & -\frac{a^2 y^2 + r^2}{y^2 - 1} & 0 \\ 0 & 0 & \left(\frac{2(y^2 - 1)a^2 mr}{a^2 y^2 + r^2} - a^2 - r^2 \right) (y^2 - 1) \end{pmatrix}$$

Lapse function and shift vector

```
In [11]: N = Sig.scalar_field(sqrt(Del / BB2), name='N')
print(N)
N.display()
```

Scalar field N on the 3-dimensional differentiable manifold Sigma

Out[11]: $N : \Sigma \longrightarrow \mathbb{R}$

$$(r, y, \phi) \longmapsto \sqrt{-\frac{a^2 - 2mr + r^2}{\frac{2(y^2 - 1)a^2 mr}{a^2 y^2 + r^2} - a^2 - r^2}}$$

```
In [12]: b = Sig.vector_field('beta', latex_name=r'\beta')
b[3] = -2*m*r*a/AA2
# unset components are zero
b.display()
```

Out[12]:

$$\beta = \left(\frac{2 amr}{2(y^2 - 1)a^2 mr - (a^2 y^2 + r^2)(a^2 + r^2)} \right) \frac{\partial}{\partial \phi}$$

Extrinsic curvature of Σ

We use the formula

$$K_{ij} = \frac{1}{2N} \mathcal{L}_\beta \gamma_{ij}$$

which is valid for any stationary spacetime:

```
In [13]: K = gam.lie_der(b) / (2*N)
          K.set_name('K')
          print(K) ; K.display()
```

Field of symmetric bilinear forms K on the 3-dimensional differentiable manifold Sigma

Out[13]:

$$\left(\frac{(a^3 m r^2 + 3 a m r^4 + (a^5 m - a^3 m r^2) y^4 - (a^5 m + 3 a m r^4) y^2) \sqrt{2 a^2 m r + a^2 r^2 + r^4} - (2 a^2 m r^3 + a^2 r^4 + r^6 + (a^6 - 2 a^4 m r + a^4 r^2) y^4 + 2 \sqrt{a^2 y^2 + r^2} \sqrt{a^2 (a^4 m r + a^4 r^2 - a^2 m r^3 + a^2 r^4) y^2})}{(a^4 m r + a^4 r^2 - a^2 m r^3 + a^2 r^4) y^2} \right)$$

Check (comparison with known formulas):

```
In [14]: Krp = a*m*(1-y^2)*(3*r^4+a^2*r^2+a^2*(r^2-a^2)*y^2) / rho2^2/sqrt(Del*B
          Krp
          B2)
```

Out[14]:

$$\frac{((a^2 - r^2) a^2 y^2 - a^2 r^2 - 3 r^4) (y^2 - 1) a m}{(a^2 y^2 + r^2)^2 \sqrt{-\left(\frac{2 (y^2 - 1) a^2 m r}{a^2 y^2 + r^2} - a^2 - r^2\right) (a^2 - 2 m r + r^2)}}$$

```
In [15]: K[1,3] - Krp
```

Out[15]: 0

```
In [16]: Kyp = 2*m*r*a^3*(1-y^2)*y*sqrt(Del)/rho2^2/sqrt(BB2)
          Kyp
```

Out[16]:

$$\frac{2 \sqrt{a^2 - 2 m r + r^2} (y^2 - 1) a^3 m r y}{\sqrt{-\frac{2 (y^2 - 1) a^2 m r}{a^2 y^2 + r^2} + a^2 + r^2} (a^2 y^2 + r^2)^2}}$$

```
In [17]: K[2,3] - Kyp
```

Out[17]: 0

For now on, we use the expressions Krp and Kyp above for $K_{r\phi}$ and K_{ry} , respectively:

```
In [18]: K1 = Sig.sym_bilin_form_field('K')
K1[1,3] = Krp
K1[2,3] = Kyp
K = K1
K.display()
```

Out[18]:

$$\begin{aligned}
K = & \left(\frac{((a^2 - r^2)a^2y^2 - a^2r^2 - 3r^4)(y^2 - 1)am}{(a^2y^2 + r^2)^2 \sqrt{-\left(\frac{2(y^2-1)a^2mr}{a^2y^2+r^2} - a^2 - r^2\right)(a^2 - 2mr + r^2)}} \right) dr \otimes d\phi \\
& + \left(-\frac{2\sqrt{a^2 - 2mr + r^2}(y^2 - 1)a^3mry}{\sqrt{-\frac{2(y^2-1)a^2mr}{a^2y^2+r^2} + a^2 + r^2}(a^2y^2 + r^2)^2}} \right) dy \otimes d\phi \\
& + \left(\frac{((a^2 - r^2)a^2y^2 - a^2r^2 - 3r^4)(y^2 - 1)am}{(a^2y^2 + r^2)^2 \sqrt{-\left(\frac{2(y^2-1)a^2mr}{a^2y^2+r^2} - a^2 - r^2\right)(a^2 - 2mr + r^2)}} \right) d\phi \otimes dr \\
& + \left(-\frac{2\sqrt{a^2 - 2mr + r^2}(y^2 - 1)a^3mry}{\sqrt{-\frac{2(y^2-1)a^2mr}{a^2y^2+r^2} + a^2 + r^2}(a^2y^2 + r^2)^2}} \right) d\phi \otimes dy
\end{aligned}$$

```
In [19]: K.display_comp()
```

Out[19]:

$$\begin{aligned}
K_{r\phi} &= \frac{((a^2-r^2)a^2y^2-a^2r^2-3r^4)(y^2-1)am}{(a^2y^2+r^2)^2 \sqrt{-\left(\frac{2(y^2-1)a^2mr}{a^2y^2+r^2} - a^2 - r^2\right)(a^2-2mr+r^2)}} \\
K_{y\phi} &= -\frac{2\sqrt{a^2-2mr+r^2}(y^2-1)a^3mry}{\sqrt{-\frac{2(y^2-1)a^2mr}{a^2y^2+r^2} + a^2 + r^2}(a^2y^2+r^2)^2} \\
K_{\phi r} &= \frac{((a^2-r^2)a^2y^2-a^2r^2-3r^4)(y^2-1)am}{(a^2y^2+r^2)^2 \sqrt{-\left(\frac{2(y^2-1)a^2mr}{a^2y^2+r^2} - a^2 - r^2\right)(a^2-2mr+r^2)}} \\
K_{\phi y} &= -\frac{2\sqrt{a^2-2mr+r^2}(y^2-1)a^3mry}{\sqrt{-\frac{2(y^2-1)a^2mr}{a^2y^2+r^2} + a^2 + r^2}(a^2y^2+r^2)^2}
\end{aligned}$$

The type-(1,1) tensor K^\sharp of components $K^i_j = \gamma^{ik} K_{kj}$:

```
In [20]: Ku = K.up(gam, 0)
print(Ku) ; Ku.display()
```

Tensor field of type (1,1) on the 3-dimensional differentiable manifold Sigma

Out[20]:

$$\left(\frac{\dots}{(a^6 y^6 + 3\dots)} \right)$$

$$+ \left(\frac{\dots}{(a^6)} \right)$$

$$\left(\frac{(a^3 m r^2 + 3 a m r^4 - (a^5 m - a^3 m r^2) y^2) \sqrt{a^2 y^2 + r}}{(2 a^2 m r^3 + a^2 r^4 + r^6 + (a^6 - 2 a^4 m r + a^4 r^2) y^4 + 2 \sqrt{2 a^2 m r + a^2 r^2 + r^4} + (a^4 - (a^4 m r + a^4 r^2 - a^2 m r^3 + a^2 r^4) y^2)} \right)$$

We may check that the hypersurface Σ is maximal, i.e. that $K^k_k = 0$:

```
In [21]: trK = Ku.trace()
print(trK)
trK.display()
```

Scalar field zero on the 3-dimensional differentiable manifold Sigma

Out[21]: $0 : \Sigma \rightarrow \mathbb{R}$
 $(r, y, \phi) \mapsto 0$

Connection and curvature

Let us call D the Levi-Civita connection associated with γ :

```
In [22]: D = gam.connection(name='D')
print(D) ; D
```

Levi-Civita connection D associated with the Riemannian metric gam on the 3-dimensional differentiable manifold Sigma

Out[22]: D

The Ricci tensor associated with γ :

```
In [23]: Ric = gam.ricci()
print(Ric) ; Ric
```

Field of symmetric bilinear forms Ric(gam) on the 3-dimensional differentiable manifold Sigma

Out[23]: Ric(γ)

```
In [24]: Ric.display_comp(only_nonredundant=True)
```

```
Out[24]: \newcommand{\Bold}[1]{\mathbf{#1}}\begin{array}{lcl} \mathrm{Ric}\left(\gamma\right)_{\{r, r, r\}}{\phantom{\{, r\}}}
```

```
In [25]: Ric[1,1]
```

```
Out[25]: 5 a^6 m^2 r^4 + 2 a^4 m^2 r^6 + 8 a^4 m r^7 - 7 a^2 m^2 r^8 + 7 a^2 m r^9 + 2 m r^11
+ (a^10 m^2 + 3 a^10 m r - 14 a^8 m^2 r^2 - 11 a^6 m^2 r^4 + 3 a^6 m r^5 + 6 (a^8 m + 2 a^6 m^3) r^3) y^6
+ (3 a^6 m - 4 a^4 m^3) r^5
- (a^10 m^2 + 9 a^10 m r - 30 a^8 m^2 r^2 - 35 a^6 m^2 r^4 - 16 a^4 m^2 r^6 + 4 a^4 m r^7 y^4
+ (17 a^6 m + 4 a^4 m^3) r^5 + 2 (11 a^8 m + 12 a^6 m^3) r^3)
- (16 a^8 m^2 r^2 + 29 a^6 m^2 r^4 + 18 a^4 m^2 r^6 + 16 a^4 m r^7 - 7 a^2 m^2 r^8 + 5 a^2 m r^9 y^2
+ (17 a^6 m - 8 a^4 m^3) r^5 + 6 (a^8 m - 2 a^6 m^3) r^3)
-----
4 a^6 m^2 r^6 + 6 a^4 m r^9 + 3 a^2 r^12 - 2 m r^13 + r^14 + (3 a^4 - 8 a^2 m^2) r^10
+ (a^6 - 4 a^4 m^2) r^8
+ (a^14 - 6 a^12 m r - 6 a^8 m r^5 + a^8 r^6 + 3 (a^10 + 4 a^8 m^2) r^4 - 4 y^8 + 4
(3 a^10 m + 2 a^8 m^3) r^3 + 3 (a^12 + 4 a^10 m^2) r^2)
(a^6 m - 2 a^4 m^3) r^7 + 4
(a^12 m r - 5 a^6 m r^7 + a^6 r^8 + (3 a^8 + 8 a^6 m^2) r^6 - (9 a^8 m + 4 a^6 m^3) r^5 y^6 + 2
+ (3 a^10 + 4 a^8 m^2) r^4 - (3 a^10 m - 4 a^8 m^3) r^3 + (a^12 - 4 a^10 m^2) r^2)
(2 a^10 m^2 r^2 + 16 a^6 m^3 r^5 - 12 a^4 m r^9 + 3 a^4 r^10 + (9 a^6 + 14 a^4 m^2) r^8 - 2 y^4 + 4
(9 a^6 m + 2 a^4 m^3) r^7 + 3 (3 a^8 - 2 a^6 m^2) r^6 + 3 (a^10 - 6 a^8 m^2) r^4 + 2
(3 a^10 m - 2 a^8 m^3) r^3)
(2 a^8 m^2 r^4 - 3 a^4 m r^9 - 3 a^2 m r^11 + a^2 r^12 + (3 a^4 + 2 a^2 m^2) r^10 + 3 y^2
(a^6 - 2 a^4 m^2) r^8 + (3 a^6 m + 4 a^4 m^3) r^7 + (a^8 - 6 a^6 m^2) r^6
+ (3 a^8 m - 4 a^6 m^3) r^5)
```

In [26]: Ric[1,2]

$$\begin{aligned}
& (3a^{10}m - 4a^8m^2r + 6a^8mr^2 - 8a^6m^2r^3 + 3a^6mr^4)y^5 + 2 \\
& (2a^8m^2r - 3a^8mr^2 + 12a^6m^2r^3 - 6a^6mr^4 + 6a^4m^2r^5 - 3a^4mr^6)y^3 \\
& - (16a^6m^2r^3 + 9a^6mr^4 + 12a^4m^2r^5 + 18a^4mr^6 + 9a^2mr^8)y \\
& \hline
& 4a^4m^2r^6 + 4a^4mr^7 + a^4r^8 + 4a^2mr^9 + 2a^2r^{10} + r^{12} \\
& + (a^{12} - 4a^{10}mr - 4a^8mr^3 + a^8r^4 + 2(a^{10} + 2a^8m^2)r^2)y^8 + 4 \\
& (a^{10}mr - 2a^8mr^3 - 3a^6mr^5 + a^6r^6 + 2(a^8 + a^6m^2)r^4 + (a^{10} - 2a^8m^2)r^2)y^6 \\
& + 2(2a^8m^2r^2 + 6a^8mr^3 - 6a^4mr^7 + 3a^4r^8 + 2(3a^6 + a^4m^2)r^6y^4 + 4 \\
& + (3a^8 - 8a^6m^2)r^4) \\
& (2a^6m^2r^4 + 3a^6mr^5 + 2a^4mr^7 + 2a^4r^8 - a^2mr^9 + a^2r^{10} + (a^6 - 2a^4m^2)r^6)y^2
\end{aligned}$$

In [27]: Ric[1,3]

Out[27]: 0

In [28]: Ric[2,2]

$$\begin{aligned}
& 6a^6m^2r^4 + 4a^4m^2r^6 + 7a^4mr^7 - 2a^2m^2r^8 + 5a^2mr^9 + mr^{11} + 2 \\
& (3a^{10}mr - 10a^8m^2r^2 - 10a^6m^2r^4 + 3a^6mr^5 + 2(3a^8m + 4a^6m^3)r^3)y^6 \\
& + (3a^6m - 8a^4m^3)r^5 \\
& - (9a^{10}mr - 34a^8m^2r^2 - 36a^6m^2r^4 - 2a^4m^2r^6 - a^4mr^7 + (7a^6m + 8a^4m^3)r^5y^4 \\
& + (17a^8m + 32a^6m^3)r^3) \\
& - 2(7a^8m^2r^2 + 11a^6m^2r^4 + 3a^4m^2r^6 + 7a^4mr^7 - a^2m^2r^8 + 2a^2mr^9 + 8y^2 \\
& (a^6m - a^4m^3)r^5 + (3a^8m - 8a^6m^3)r^3) \\
& \hline
& 4a^4m^2r^6 + 4a^4mr^7 + a^4r^8 + 4a^2mr^9 + 2a^2r^{10} + r^{12} \\
& - (a^{12} - 4a^{10}mr - 4a^8mr^3 + a^8r^4 + 2(a^{10} + 2a^8m^2)r^2)y^{10} \\
& + (a^{12} - 8a^{10}mr + 4a^8mr^3 + 12a^6mr^5 - 4a^6r^6 - (7a^8 + 8a^6m^2)r^4 - 2y^8 \\
& (a^{10} - 6a^8m^2)r^2) \\
& + 2(2a^{10}mr - 10a^8mr^3 - 6a^6mr^5 + 6a^4mr^7 - 3a^4r^8 - 2(2a^6 + a^4m^2)r^6y^6 \\
& + (a^8 + 12a^6m^2)r^4 + 2(a^{10} - 3a^8m^2)r^2) \\
& + 2(2a^8m^2r^2 + 6a^8mr^3 - 6a^6mr^5 - 10a^4mr^7 - a^4r^8 + 2a^2mr^9 - 2a^2r^{10} + 2y^4 \\
& (2a^6 + 3a^4m^2)r^6 + 3(a^8 - 4a^6m^2)r^4) \\
& + (8a^6m^2r^4 + 12a^6mr^5 + 4a^4mr^7 + 7a^4r^8 - 8a^2mr^9 + 2a^2r^{10} - r^{12} + 4y^2 \\
& (a^6 - 3a^4m^2)r^6)
\end{aligned}$$

In [29]: Ric[2,3]

Out[29]: 0

In [30]: `Ric[3,3]`

$$\begin{aligned}
 \text{Out[30]: } & a^6 m^2 r^4 + 4 a^4 m^3 r^5 + 10 a^4 m^2 r^6 + a^4 m r^7 + 13 a^2 m^2 r^8 + 2 a^2 m r^9 + m r^{11} \\
 & + (a^{10} m^2 + 3 a^{10} m r - 18 a^8 m^2 r^2 - 15 a^6 m^2 r^4 + 3 a^6 m r^5 + 2 y^8 \\
 & \quad (3 a^8 m + 10 a^6 m^3) r^3) \\
 & - (2 a^{10} m^2 + 3 a^{10} m r - 38 a^8 m^2 r^2 - 22 a^6 m^2 r^4 + 2 a^4 m^2 r^6 - 5 a^4 m r^7 y^6 \\
 & \quad - (7 a^6 m - 4 a^4 m^3) r^5 + (a^8 m + 60 a^6 m^3) r^3) \\
 & + (a^{10} m^2 - 22 a^8 m^2 r^2 + 2 a^6 m^2 r^4 + 14 a^4 m^2 r^6 - 3 a^4 m r^7 + 13 a^2 m^2 r^8 + a^2 m r^9 y^4 \\
 & \quad - 3 (3 a^6 m - 4 a^4 m^3) r^5 - 5 (a^8 m - 12 a^6 m^3) r^3) \\
 & + (2 a^8 m^2 r^2 - 20 a^6 m^3 r^3 - 10 a^6 m^2 r^4 - 22 a^4 m^2 r^6 - 3 a^4 m r^7 - 26 a^2 m^2 r^8 - 3 y^2 \\
 & \quad a^2 m r^9 - m r^{11} - (a^6 m + 12 a^4 m^3) r^5) \\
 & \hline
 & 2 a^2 m r^9 + a^2 r^{10} + r^{12} + (a^{12} - 2 a^{10} m r + a^{10} r^2) y^{10} \\
 & + (2 a^{10} m r + 5 a^{10} r^2 - 8 a^8 m r^3 + 5 a^8 r^4) y^8 + 2 \\
 & \quad (4 a^8 m r^3 + 5 a^8 r^4 - 6 a^6 m r^5 + 5 a^6 r^6) y^6 + 2 \\
 & \quad (6 a^6 m r^5 + 5 a^6 r^6 - 4 a^4 m r^7 + 5 a^4 r^8) y^4 \\
 & \quad + (8 a^4 m r^7 + 5 a^4 r^8 - 2 a^2 m r^9 + 5 a^2 r^{10}) y^2
 \end{aligned}$$

The scalar curvature $R = \gamma^{ij} R_{ij}$:

In [31]: `R = gam.ricci_scalar(name='R')`
`print(R)`
`R.display()`

Scalar field R on the 3-dimensional differentiable manifold Sigma

$$\begin{aligned}
 \text{Out[31]: } & r(\gamma) : \Sigma \longrightarrow \mathbb{R} \\
 & (r, y, \phi) \longmapsto \frac{2(a^6 m^2 r^4 + 6 a^4 m^2 r^6 + 9 a^2 m^2 r^8 - (a^{10} m^2 - 6 a^8 m^2 r^2 + 8 a^6 m^3 r^3 - 3 a^6 m^2 r^4) y^6 \\
 & \quad + (a^{10} m^2 - 8 a^8 m^2 r^2 + 16 a^6 m^3 r^3 - 3 a^6 m^2 r^4 - 6 a^4 m^2 r^6) y^4 + (2 a^8 m^2 r^2 - 8 a^6 m^3 r^3 \\
 & \quad + 4 a^{12} m r - 12 a^{10} m r^3 - 16 a^8 m r^5 + 5 a^8 r^6 + 2(5 a^{10} + 6 a^8 m^2) r^4 + (5 a^{12} - 8 a^{10} m^2) r \\
 & \quad (2 a^{10} m^2 r^2 + 8 a^{10} m r^3 - 4 a^8 m r^5 - 12 a^6 m r^7 + 5 a^6 r^8 + 2(5 a^8 + 3 a^6 m^2) r^6 + (5 a^{10} - 1 \\
 & \quad (6 a^8 m^2 r^4 + 12 a^8 m r^5 + 4 a^6 m r^7 - 8 a^4 m r^9 + 5 a^4 r^{10} + 2(5 a^6 + a^4 m^2) r^8 + (5 a^8 - 12 a^6 \\
 & \quad + (12 a^6 m^2 r^6 + 16 a^6 m r^7 + 12 a^4 m r^9 + 10 a^4 r^{10} - 4 a^2 m r^{11} + 5 a^2 r^{12} + (5 a^6 - 8 a^4 m^2) r
 \end{aligned}$$

3+1 Einstein equations

Let us check that the vacuum 3+1 Einstein equations are satisfied.

We start by the constraint equations:

Hamiltonian constraint

Let us first evaluate the term $K_{ij} K^{ij}$:

```
In [32]: Kuu = Ku.up(gam, 1)
trKK = K['_ij']*Kuu['^ij']
print(trKK) ; trKK.display()
```

Scalar field on the 3-dimensional differentiable manifold Sigma

```
Out[32]:  Σ      → ℝ
          (r, y, φ) ↦
          2 (a^6 m^2 r^4 + 6 a^4 m^2 r^6 + 9 a^2 m^2 r^8 - (a^10 m^2 - 6 a^8 m^2 r^2 + 8 a^6 m^3 r^3 - 3 a^6 m^2 r^4) y^6
          + (a^10 m^2 - 8 a^8 m^2 r^2 + 16 a^6 m^3 r^3 - 3 a^6 m^2 r^4 - 6 a^4 m^2 r^6) y^4 + (2 a^8 m^2 r^2 - 8 a^6 m^3 r^3 - a^6 m^2 r
          4 a^4 m^2 r^8 + 4 a^4 m r^9 + a^4 r^10 + 4 a^2 m r^11 + 2 a^2 r^12 + r^14 + (a^14 - 4 a^12 m r - 4 a^10 m r^3 + a^10 r^4 + 2 (a^1
          + (4 a^12 m r - 12 a^10 m r^3 - 16 a^8 m r^5 + 5 a^8 r^6 + 2 (5 a^10 + 6 a^8 m^2) r^4 + (5 a^12 - 8 a^10 m^2) r^2) y^8 + 2
          (2 a^10 m^2 r^2 + 8 a^10 m r^3 - 4 a^8 m r^5 - 12 a^6 m r^7 + 5 a^6 r^8 + 2 (5 a^8 + 3 a^6 m^2) r^6 + (5 a^10 - 12 a^8 m^2)
          (6 a^8 m^2 r^4 + 12 a^8 m r^5 + 4 a^6 m r^7 - 8 a^4 m r^9 + 5 a^4 r^10 + 2 (5 a^6 + a^4 m^2) r^8 + (5 a^8 - 12 a^6 m^2) r^6)
          + (12 a^6 m^2 r^6 + 16 a^6 m r^7 + 12 a^4 m r^9 + 10 a^4 r^10 - 4 a^2 m r^11 + 5 a^2 r^12 + (5 a^6 - 8 a^4 m^2) r^8) y^2
```

The vacuum Hamiltonian constraint equation is

$$R + K^2 - K_{ij}K^{ij} = 0$$

```
In [33]: Ham = R + trK^2 - trKK
print(Ham) ; Ham.display()
```

Scalar field zero on the 3-dimensional differentiable manifold Sigma

```
Out[33]: 0: Σ      → ℝ
          (r, y, φ) ↦ 0
```

Momentum constraint

In vacuum, the momentum constraint is

$$D_j K^j_i - D_i K = 0$$

```
In [34]: mom = D(Ku).trace(0,2) - D(trK)
print(mom)
mom.display()
```

1-form on the 3-dimensional differentiable manifold Sigma

```
Out[34]: 0
```

Dynamical Einstein equations

Let us first evaluate the symmetric bilinear form $k_{ij} := K_{ik}K^k_j$:

```
In [35]: KK = K['_ik']*Ku['^k_j']
print(KK)
```

Tensor field of type (0,2) on the 3-dimensional differentiable manifold Sigma

```
In [36]: KK1 = KK.symmetrize()
KK == KK1
```

```
Out[36]: True
```

```
In [37]: KK = KK1
print(KK)
```

Field of symmetric bilinear forms on the 3-dimensional differentiable manifold Sigma

```
In [38]: KK.set_name('(KK)')
KK.display_comp()
```

```
Out[38]:
```

$$\begin{aligned}
(KK)_{rr} &= \frac{a^6 m^2 r^4 + 6 a^4 m^2 r^6 + 9 a^2 m^2 r^8 - (a^{10} m^2 - 2 a^8 m^2 r^2 + a^6 m^2 r^4) y^6 + (a^{10} m^2 + 5 a^6 m^2 r^4 - 6 a^4 m^2 r^6 - 2 a^8 m^2 r^2 + 5 a^6 m^2 r^4 + 9 a^2 m^2 r^8) y^2}{4 a^6 m^2 r^6 + 6 a^4 m r^9 + 3 a^2 r^{12} - 2 m r^{13} + r^{14} + (3 a^4 - 8 a^2 m^2) r^{10} + (a^6 - 4 a^4 m^2) r^8 + (a^{14} - 6 a^{12} m r - 6 a^8 m r^5 + a^8 r^6 + 3 (a^{10} + 4 a^8 m^2) r^4 - 4 (3 a^{10} m + 2 a^8 m^3) r^3 + 3 (a^{12} + 4 a^{10} m^2) r^2 + (a^6 m - 2 a^4 m^3) r^7 + 4 (a^{12} m r - 5 a^6 m r^7 + a^6 r^8 + (3 a^8 + 8 a^6 m^2) r^6 - (9 a^8 m + 4 a^6 m^3) r^5 + (3 a^{10} + 4 a^8 m^2) r^4 - (3 a^{10} m - 4 + (a^{12} - 4 a^{10} m^2) r^2) (2 a^{10} m^2 r^2 + 16 a^6 m^3 r^5 - 12 a^4 m r^9 + 3 a^4 r^{10} + (9 a^6 + 14 a^4 m^2) r^8 - 2 (9 a^6 m + 2 a^4 m^3) r^7 + 3 (3 a^8 (a^{10} - 6 a^8 m^2) r^4 + 2 (3 a^{10} m - 2 a^8 m^3) r^3) + 4 (2 a^8 m^2 r^4 - 3 a^4 m r^9 - 3 a^2 m r^{11} + a^2 r^{12} + (3 a^4 + 2 a^2 m^2) r^{10} + 3 (a^6 - 2 a^4 m^2) r^8 + (3 a^6 m + 4 a^4 m^2) r^6 + (a^8 - 6 a^6 m^2) r^6 + (3 a^8 m - 4 a^6 m^3) r^5)} \\
(KK)_{ry} &= \frac{2 ((a^8 m^2 r - a^6 m^2 r^3) y^5 - (a^8 m^2 r + 3 a^4 m^2 r^5) y^3 + (a^6 m^2 r^3 + 3 a^4 m^2 r^5) y)}{4 a^4 m^2 r^6 + 4 a^4 m r^7 + a^4 r^8 + 4 a^2 m r^9 + 2 a^2 r^{10} + r^{12} + (a^{12} - 4 a^{10} m r - 4 a^8 m r^3 + a^8 r^4 + 2 (a^{10} + 2 a^8 m (a^{10} m r - 2 a^8 m r^3 - 3 a^6 m r^5 + a^6 r^6 + 2 (a^8 + a^6 m^2) r^4 + (a^{10} - 2 a^8 m^2) r^2) y^6 + 2 (2 a^8 m^2 r^2 + 6 a^8 m r^3 - 6 a^4 m r^7 + 3 a^4 r^8 + 2 (3 a^6 + a^4 m^2) r^6 + (3 a^8 - 8 a^6 m^2) r^4) y^4 + 4 (2 a^6 m^2 r^4 + 3 a^6 m r^5 + 2 a^4 m r^7 + 2 a^4 r^8 - a^2 m r^9 + a^2 r^{10} + (a^6 - 2 a^4 m^2) r^6) y^2} \\
(KK)_{yy} &= \frac{2 ((a^8 m^2 r - a^6 m^2 r^3) y^5 - (a^8 m^2 r + 3 a^4 m^2 r^5) y^3 + (a^6 m^2 r^3 + 3 a^4 m^2 r^5) y)}{4 a^4 m^2 r^6 + 4 a^4 m r^7 + a^4 r^8 + 4 a^2 m r^9 + 2 a^2 r^{10} + r^{12} + (a^{12} - 4 a^{10} m r - 4 a^8 m r^3 + a^8 r^4 + 2 (a^{10} + 2 a^8 m (a^{10} m r - 2 a^8 m r^3 - 3 a^6 m r^5 + a^6 r^6 + 2 (a^8 + a^6 m^2) r^4 + (a^{10} - 2 a^8 m^2) r^2) y^6 + 2 (2 a^8 m^2 r^2 + 6 a^8 m r^3 - 6 a^4 m r^7 + 3 a^4 r^8 + 2 (3 a^6 + a^4 m^2) r^6 + (3 a^8 - 8 a^6 m^2) r^4) y^4 + 4 (2 a^6 m^2 r^4 + 3 a^6 m r^5 + 2 a^4 m r^7 + 2 a^4 r^8 - a^2 m r^9 + a^2 r^{10} + (a^6 - 2 a^4 m^2) r^6) y^2} \\
(KK)_{y\phi} &= -\frac{4 ((a^8 m^2 r^2 - 2 a^6 m^3 r^3 + a^6 m^2 r^4) y^4 - (a^8 m^2 r^2 - 2 a^6 m^3 r^3 + a^6 m^2 r^4) y^2)}{4 a^4 m^2 r^6 + 4 a^4 m r^7 + a^4 r^8 + 4 a^2 m r^9 + 2 a^2 r^{10} + r^{12} + (a^{12} - 4 a^{10} m r - 4 a^8 m r^3 + a^8 r^4 + 2 (a^{10} + 2 a^8 m (a^{10} m r - 2 a^8 m r^3 - 3 a^6 m r^5 + a^6 r^6 + 2 (a^8 + a^6 m^2) r^4 + (a^{10} - 2 a^8 m^2) r^2) y^6 + 2 (2 a^8 m^2 r^2 + 6 a^8 m r^3 - 6 a^4 m r^7 + 3 a^4 r^8 + 2 (3 a^6 + a^4 m^2) r^6 + (3 a^8 - 8 a^6 m^2) r^4) y^4 + 4 (2 a^6 m^2 r^4 + 3 a^6 m r^5 + 2 a^4 m r^7 + 2 a^4 r^8 - a^2 m r^9 + a^2 r^{10} + (a^6 - 2 a^4 m^2) r^6) y^2} \\
(KK)_{\phi\phi} &= \frac{a^6 m^2 r^4 + 6 a^4 m^2 r^6 + 9 a^2 m^2 r^8 + (a^{10} m^2 - 6 a^8 m^2 r^2 + 8 a^6 m^3 r^3 - 3 a^6 m^2 r^4) y^8 - 2 (a^{10} m^2 - 7 a^8 m^2 r^2 + 12 a^6 m^3 r^3 - 3 a^6 m^2 r^4 - 3 a^4 m^2 r^6) y^6 + (a^{10} m^2 - 10 a^8 m^2 r^2 + 24 a^6 m^3 r^3 - 2 a^6 m^2 r^4 - 6 a^4 m^2 r^6 + 9 a^2 m^2 r^8) y^4 + 2 (a^8 m^2 r^2 - 4 a^6 m^3 r^3 - a^6 m^2 r^4 - 3 a^4 m^2 r^6 - 9 a^2 m^2 r^8) y^2}{2 a^2 m r^9 + a^2 r^{10} + r^{12} + (a^{12} - 2 a^{10} m r + a^{10} r^2) y^{10} + (2 a^{10} m r + 5 a^{10} r^2 - 8 a^8 m r^3 + 5 a^8 r^4) y^8 + 2 (4 a^8 m r^3 + 5 a^8 r^4 - 6 a^6 m r^5 + 5 a^6 r^6) y^6 + 2 (6 a^6 m r^5 + 5 a^6 r^6 - 4 a^4 m r^7 + 5 a^4 r^8) y^4 + (8 a^4 m r^7 + 5 a^4 r^8 - 2 a^2 m r^9 + 5 a^2 r^{10}) y^2}
\end{aligned}$$

In vacuum and for stationary spacetimes, the dynamical Einstein equations are

$$\mathcal{L}_\beta K_{ij} - D_i D_j N + N (R_{ij} + KK_{ij} - 2K_{ik} K_j^k) = 0$$

```
In [39]: dyn = K.lie_der(b) - D(D(N)) + N*( Ric + trK*K - 2*KK )
print(dyn)
dyn.display()
```

Tensor field of type (0,2) on the 3-dimensional differentiable manifold Sigma

```
Out[39]: 0
```

Electric and magnetic parts of the Weyl tensor

The **electric part** is the bilinear form E given by

$$E_{ij} = R_{ij} + KK_{ij} - K_{ik}K^k_j$$

```
In [40]: E = Ric + trK*K - KK
print(E)
```

Field of symmetric bilinear forms +Ric(gam)-(KK) on the 3-dimensional differentiable manifold Sigma

```
In [41]: E.set_name('E')
E.display_comp(only_nonzero=False)
```

```
Out[41]:
```

$$E_{rr} = \frac{3a^4mr^3 - 2a^2m^2r^4 + 5a^2mr^5 + 2mr^7 + 3(a^6mr - 2a^4m^2r^2 + a^4mr^3)y^4 - (9a^6mr - 6a^4m^2r^2 + 16a^4mr^3 - 2a^2m^2r^4 + 7a^2mr^5)y^2}{2a^4mr^5 + 2a^2r^8 - 2mr^9 + r^{10} + (a^4 - 4a^2m^2)r^6 + (a^{10} - 4a^8mr - 4a^6mr^3 + a^6r^4 + 2(a^8 + 2a^6m^2)r^2)y^6 + (2a^8mr - 8a^6mr^3 - 10a^4mr^5 + 3a^4r^6 + 2(3a^6 + 4a^4m^2)r^4 + (3a^8 - 4a^6m^2)r^2)y^4 + (4a^6mr^3 - 4a^4mr^5 - 8a^2mr^7 + 3a^2r^8 + 2(3a^4 + 2a^2m^2)r^6 + (3a^6 - 8a^4m^2)r^4)y^2}$$

$$E_{ry} = \frac{3((a^6m + a^4mr^2)y^3 - 3(a^4mr^2 + a^2mr^4)y)}{2a^2mr^5 + a^2r^6 + r^8 + (a^8 - 2a^6mr + a^6r^2)y^6 + (2a^6mr + 3a^6r^2 - 4a^4mr^3 + 3a^4r^4)y^4 + (4a^4mr^3 + 3a^4r^4 - 2a^2mr^5 + 3a^2r^6)y^2}$$

$$E_{r\phi} = 0$$

$$E_{yr} = \frac{3((a^6m + a^4mr^2)y^3 - 3(a^4mr^2 + a^2mr^4)y)}{2a^2mr^5 + a^2r^6 + r^8 + (a^8 - 2a^6mr + a^6r^2)y^6 + (2a^6mr + 3a^6r^2 - 4a^4mr^3 + 3a^4r^4)y^4 + (4a^4mr^3 + 3a^4r^4 - 2a^2mr^5 + 3a^2r^6)y^2}$$

$$E_{yy} = \frac{3a^4mr^3 - 4a^2m^2r^4 + 4a^2mr^5 + mr^7 + 6(a^6mr - 2a^4m^2r^2 + a^4mr^3)y^4 - (9a^6mr - 12a^4m^2r^2 + 14a^4mr^3 - 4a^2m^2r^4 + 5a^2mr^5)y^2}{(a^8 - 2a^6mr + a^6r^2)y^8 - 2a^2mr^5 - a^2r^6 - r^8 - (a^8 - 4a^6mr - 2a^6r^2 + 4a^4mr^3 - 3a^4r^4)y^6 - (2a^6mr + 3a^6r^2 - 8a^4mr^3 + 2a^2mr^5 - 3a^2r^6)y^4 - (4a^4mr^3 + 3a^4r^4 - 4a^2mr^5 + 2a^2r^6 - r^8)y^2}$$

$$E_{y\phi} = 0$$

$$E_{\phi r} = 0$$

$$E_{\phi y} = 0$$

$$E_{\phi\phi} = \frac{2a^2m^2r^4 + a^2mr^5 + mr^7 + 3(a^6mr - 2a^4m^2r^2 + a^4mr^3)y^6 - (3a^6mr - 12a^4m^2r^2 + a^4mr^3 - 2a^2m^2r^4 - 2a^2 - (6a^4m^2r^2 + 2a^4mr^3 + 4a^2m^2r^4 + 3a^2mr^5 + mr^7)y^2}{a^8y^8 + 4a^6r^2y^6 + 6a^4r^4y^4 + 4a^2r^6y^2 + r^8}$$

The **magnetic part** is the bilinear form B defined by

$$B_{ij} = \epsilon^k_{il} D_k K^l_j,$$

where ϵ^k_{il} are the components of the type-(1,2) tensor ϵ^\sharp , related to the Levi-Civita alternating tensor ϵ associated with γ by $\epsilon^k_{il} = \gamma^{km} \epsilon_{mil}$. In SageManifolds, ϵ is obtained by the command `volume_form()` and ϵ^\sharp by the command `volume_form(1)` (1 = one index raised):

```
In [42]: eps = gam.volume_form()
print(eps) ; eps.display()
```

3-form eps_gam on the 3-dimensional differentiable manifold Sigma

Out[42]:

$$\epsilon_\gamma = \left(\frac{\sqrt{2a^2mr + a^2r^2 + r^4 + (a^4 - 2a^2mr + a^2r^2)y^2} \sqrt{a^2y^2 + r^2}}{\sqrt{a^2 - 2mr + r^2}} \right) dr \wedge dy \wedge d\phi$$

```
In [43]: epsu = gam.volume_form(1)
print(epsu) ; epsu.display()
```

Tensor field of type (1,2) on the 3-dimensional differentiable manifold Sigma

Out[43]:

$$\begin{aligned} & \left(\frac{\sqrt{2a^2mr + a^2r^2 + r^4 + (a^4 - 2a^2mr + a^2r^2)y^2} \sqrt{a^2 - 2mr + r^2}}{\sqrt{a^2y^2 + r^2}} \right) \frac{\partial}{\partial r} \\ & \otimes d\phi + \left(-\frac{\sqrt{2a^2mr + a^2r^2 + r^4 + (a^4 - 2a^2mr + a^2r^2)y^2} \sqrt{a^2 - 2mr + r^2}}{\sqrt{a^2y^2 + r^2}} \right) \\ & \otimes d\phi \otimes dy + \left(\frac{\sqrt{2a^2mr + a^2r^2 + r^4 + (a^4 - 2a^2mr + a^2r^2)y^2} (y^2 - 1)}{\sqrt{a^2y^2 + r^2} \sqrt{a^2 - 2mr + r^2}} \right) \\ & \otimes dr \otimes d\phi + \left(-\frac{\sqrt{2a^2mr + a^2r^2 + r^4 + (a^4 - 2a^2mr + a^2r^2)y^2} (y^2 - 1)}{\sqrt{a^2y^2 + r^2} \sqrt{a^2 - 2mr + r^2}} \right) \\ & \otimes d\phi \otimes dr \\ & + \left(-\frac{\sqrt{2a^2mr + a^2r^2 + r^4 + (a^4 - 2a^2mr + a^2r^2)y^2} (a^2y^2 + r^2)^{\frac{3}{2}}}{((a^4 - 2a^2mr + a^2r^2)y^4 - 2a^2mr - a^2r^2 - r^4 - (a^4 - 4a^2mr - r^4)y^2) \sqrt{a^2 - 2mr + r^2}} \right) \\ & \otimes dr \otimes dy \\ & + \left(\frac{\sqrt{2a^2mr + a^2r^2 + r^4 + (a^4 - 2a^2mr + a^2r^2)y^2} (a^2y^2 + r^2)^{\frac{3}{2}}}{((a^4 - 2a^2mr + a^2r^2)y^4 - 2a^2mr - a^2r^2 - r^4 - (a^4 - 4a^2mr - r^4)y^2) \sqrt{a^2 - 2mr + r^2}} \right) \\ & \otimes dy \otimes dr \end{aligned}$$

```
In [44]: DKu = D(Ku)
B = epsu['^k_il']*DKu['^l_jk']
print(B)
```

Tensor field of type (0,2) on the 3-dimensional differentiable manifold Sigma

Let us check that B is symmetric:

```
In [45]: B1 = B.symmetrize()
B == B1
```

Out[45]: True

Accordingly, we set

```
In [46]: B = B1
          B.set_name('B')
          print(B)
```

Field of symmetric bilinear forms B on the 3-dimensional differentiable manifold Sigma

```
In [47]: B.display_comp(only_nonzero=False)
```

```
Out[47]:
```

$$\begin{aligned}
 B_{rr} &= \frac{(a^7m-2a^5m^2r+a^5mr^2)y^5-(3a^7m-2a^5m^2r+8a^5mr^2-6a^3m^2r^3+5a^3mr^4)y^3+3(3a^5mr^2-2a^3m^2r^3+5a^3mr^4+2amr^6)y}{2a^4mr^5+2a^2r^8-2mr^9+r^{10}+(a^4-4a^2m^2)r^6+(a^{10}-4a^8mr-4a^6mr^3+a^6r^4+2(a^8+2a^6m^2)r^2)y^6+(2a^8mr-8a^6mr^3-10a^4mr^5+3a^4r^6+2(3a^6+4a^4m^2)r^4+(3a^8-4a^6m^2)r^2)y^4+(4a^6mr^3-4a^4mr^5-8a^2mr^7+3a^2r^8+2(3a^4+2a^2m^2)r^6+(3a^6-8a^4m^2)r^4)y^2} \\
 B_{ry} &= -\frac{3(a^3mr^3+amr^5-3(a^5mr+a^3mr^3)y^2)}{2a^2mr^5+a^2r^6+r^8+(a^8-2a^6mr+a^6r^2)y^6+(2a^6mr+3a^6r^2-4a^4mr^3+3a^4r^4)y^4+(4a^4mr^3+3a^4r^4-2a^2mr^5+3a^2r^6)y^2} \\
 B_{r\phi} &= 0 \\
 B_{yr} &= -\frac{3(a^3mr^3+amr^5-3(a^5mr+a^3mr^3)y^2)}{2a^2mr^5+a^2r^6+r^8+(a^8-2a^6mr+a^6r^2)y^6+(2a^6mr+3a^6r^2-4a^4mr^3+3a^4r^4)y^4+(4a^4mr^3+3a^4r^4-2a^2mr^5+3a^2r^6)y^2} \\
 B_{yy} &= \frac{2(a^7m-2a^5m^2r+a^5mr^2)y^5-(3a^7m-4a^5m^2r+10a^5mr^2-12a^3m^2r^3+7a^3mr^4)y^3+3(3a^5mr^2-4a^3m^2r^3+4a^3mr^4+amr^6)y}{(a^8-2a^6mr+a^6r^2)y^8-2a^2mr^5-a^2r^6-r^8-(a^8-4a^6mr-2a^6r^2+4a^4mr^3-3a^4r^4)y^6-(2a^6mr+3a^6r^2-8a^4mr^3+2a^2mr^5-3a^2r^6)y^4-(4a^4mr^3+3a^4r^4-4a^2mr^5+2a^2r^6-r^8)y^2} \\
 B_{y\phi} &= 0 \\
 B_{\phi r} &= 0 \\
 B_{\phi y} &= 0 \\
 B_{\phi\phi} &= -\frac{(a^7m-2a^5m^2r+a^5mr^2)y^7-(a^7m-4a^5m^2r+3a^5mr^2-6a^3m^2r^3+2a^3mr^4)y^5-(2a^5m^2r-2a^5mr^2+12a^3m^2r^3+a^3mr^4+3amr^6)y^3+3(2a^3m^2r^3+a^3mr^4+amr^6)y}{a^8y^8+4a^6r^2y^6+6a^4r^4y^4+4a^2r^6y^2+r^8}
 \end{aligned}$$